Test 4 solutions (white version)

1. (15 points) Find the volume of the tetrahedron with

$0 < x < 3$
$0 < y < 1,$
$0 < z$
$x/3 + y - 4z > 1$

Note: If it is projected into the x-y plane, it becomes a triangle in the first quadrant, bounded by x=3, y=1, and y=-x/3 +1 . We can write the volume as a two variable integral over this triangle (for example). The height at point (x,y) will be x/12 + y/4 - 1/4

An explicit integral for the answer is:

```
> Int(Int(x/12 + y/4 - 1/4, y=1-x/3..1), x=0..3);
```

$$\int_{0}^{3}\int_{1-\frac{1}{3}x}^{1} \frac{1}{12}x + \frac{1}{4}y - \frac{1}{4}\,\mathrm{d}y\,\mathrm{d}x \qquad (1)$$

The value of the integral is:

```
> int(x/12 + y/4 - 1/4, y=1-x/3..1);
```

$$\frac{1}{36}x^2 + \frac{1}{8} - \frac{1}{8}\left(1 - \frac{1}{3}x\right)^2 - \frac{1}{12}x \qquad (2)$$

```
> int(%, x=0..3);
```

$$\frac{1}{8} \qquad (3)$$

2. (15 points) Determine whether

h := (2 x -2 y^2)i + (2 y + 2 x^2) j

is a gradient, then calculate the line integral of h over the curve,

C := the circular arc x^2 + y^2 = 25, y ≥ 0 from (-5,0) to (0,5)

a and b) Is h a gradient? Let's do the calculation:

```
> P := (x,y) -> 2 * x -2* y^2 : Q := (x,y) -> 2*y +2* x^2 :
> diff(P(x,y),y) - diff(Q(x,y),x);
```

(4)

$$-4\,y-4\,x \qquad (4)$$

This is not zero. The answer is NO. Therefore we will need to parametrize the curve, which can be done in many ways, for example, x = 5 cos(t), y = 5 sin(t), t from pi to pi/2

c) Evaluate the line integral:

```
> int((2*5*cos(t)-2*(5*sin(t))^2)*(-5*sin(t)) + (2*5*sin(t) +2* (5*
cos(t))^2)*(5*cos(t)), t=Pi..Pi/2);
```

$$0 \qquad (5)$$

3. Evaluate the integral of z over the region Omega that lies

i) above the x-y plane
ii) outside the sphere of radius 2, centered at the origin
iii) below the cone z^2 = x^2 + y^2
iv) inside the vertical circular cylinder (like a tin-can sitting upright on the table) of radius 2, centered at the origin

a) Set the integral up explicitly in any of the 3-dimensional coordinate systems we have used in class. (You choose which one):

Cartesian is the worst choice, because the x-y integration region is a circular disk with another circular disk removed. You could write the intergral as a difference:

```
> Int(Int(Int(z, z=sqrt(4-x^2-y^2)..sqrt(x^2+y^2)),x=-sqrt(4-y^2)..
sqrt(4-y^2)), y=-2..2) - Int(Int(Int(z, z=sqrt(4-x^2-y^2)..sqrt
(x^2+y^2)),x=-sqrt(2-y^2)..sqrt(2-y^2)), y=-sqrt(2)..sqrt(2));
>
```

$$\int_{-2}^{2}\int_{-\sqrt{4-y^2}}^{\sqrt{4-y^2}}\int_{\sqrt{4-x^2-y^2}}^{\sqrt{x^2+y^2}} z\,dz\,dx\,dy-\int_{-\sqrt{2}}^{\sqrt{2}}\int_{-\sqrt{2-y^2}}^{\sqrt{2-y^2}}\int_{\sqrt{4-x^2-y^2}}^{\sqrt{x^2+y^2}} z\,dz\,dx\,dy \qquad (6)$$

In cylindrical it is:

```
> Int(Int(Int(z*r, z=sqrt(4-r^2)..r), r=sqrt(2)..2), theta=0..2*Pi);
```

$$\int_{0}^{2\pi}\int_{\sqrt{2}}^{2}\int_{\sqrt{4-r^2}}^{r} z\,r\,dz\,dr\,d\theta \qquad (7)$$

To evaluate this one:

```
> 2*Pi*int(z*r, z=sqrt(4-r^2)..r);
```

$$\pi\,r\left(2\,r^2-4\right) \qquad (8)$$

```
> int(%, r=sqrt(2)..2);
```

$$2\,\pi \qquad (9)$$

In spherical it is:

```
> Int(Int(Int((rho*cos(phi))*rho^2*sin(phi), rho=2..2/sin(phi)),
phi=Pi/4..Pi/2), theta=0..2*Pi);
```

(10)

$$\int_{0}^{2\pi}\int_{\frac{1}{4}\pi}^{\frac{1}{2}\pi}\int_{2}^{\frac{2\cdot 1}{\sin(\phi)}} \rho^3 \cos(\phi)\sin(\phi)\, d\rho\, d\phi\, d\theta \tag{10}$$

The evaluation had better be the same:

```
> 2*Pi*int((rho*cos(phi))*rho^2*sin(phi), rho=2..2/sin(phi));
```

$$\frac{1}{2}\pi\cos(\phi)\sin(\phi)\left(\frac{16\cdot 1}{\sin(\phi)^4}-16\right) \tag{11}$$

```
> int(%, phi=Pi/4..Pi/2);
```

$$2\pi \tag{12}$$

4. (15 points) Consider the region such that

x > 0, y > 0

4 < x^2 y < 6

x/4 < y < x

a) Sketch the region here:

```
> plot({4/x^2,6/x^2,x/4,x}, x=1..4);
```



b) Find a variable transformation to "straighten" the bounding curves:

```
> u := (x,y) -> y*x^2; v := (x,y) -> y/x;
```

$$u := (x, y) \rightarrow y\, x^2 \tag{13}$$

$$v := (x, y) \rightarrow \frac{y}{x}$$

c) Express x and y in terms of the new variables;

The inverse functions are

```
> x := (u,v) -> u^(1/3)/v^(1/3); y := (u,v) -> (u*v^2)^(1/3);
```

$$x := (u, v) \to \frac{u^{(1/3)}}{v^{(1/3)}} \tag{14}$$

$$y := (u, v) \to \left(u\, v^2\right)^{(1/3)}$$

d=e) Express the area as an explicit integral in u,v

```
> with(linalg):
> det([[diff(x(u,v),u),diff(x(u,v),v)],[diff(y(u,v),u),diff(y(u,v),v)]]);
```

$$\frac{1}{3}\,\frac{u^{(1/3)}\,v^{(2/3)}}{\left(u\,v^2\right)^{(2/3)}} \tag{15}$$

```
> J := (u,v) -> (1/3)*u^(-1/3) *v^(-2/3);
```

$$J := (u, v) \to \frac{1}{3}\,\frac{1}{u^{(1/3)}\,v^{(2/3)}} \tag{16}$$

The area is

```
> Int(Int(J(u,v), u=4..6), v=1/4..1);
```

$$\int_{\frac{1}{4}}^{1}\int_{4}^{6} \frac{1}{3}\,\frac{1}{u^{(1/3)}\,v^{(2/3)}}\,\mathrm{d}u\,\mathrm{d}v \tag{17}$$

```
> int(J(u,v), u=4..6);
```

$$\frac{2}{3}\,\frac{2^{(1/3)}\left(\frac{3}{4}\,3^{(2/3)}\,2^{(1/3)} - \frac{3}{2}\right)}{v^{(2/3)}} \tag{18}$$

```
> int(%, v=1/4..1);
```

$$\frac{1}{2}\,3^{(2/3)}\left(3\,2^{(2/3)} - 3\right) - \frac{1}{2}\,2^{(2/3)}\left(3\,2^{(2/3)} - 3\right) \tag{19}$$

```
> simplify(%); evalf(%);
```

$$\frac{3}{2}\,3^{(2/3)}\,2^{(2/3)} - 3\,2^{(1/3)} - \frac{3}{2}\,3^{(2/3)} + \frac{3}{2}\,2^{(2/3)} \tag{20}$$

$$0.434103568$$

```
> int(sin(t)^3, t); int(cos(t)^3,t);
```

$$-\frac{1}{3}\sin(t)^2\cos(t) - \frac{2}{3}\cos(t) \tag{21}$$

$$\frac{1}{3}\cos(t)^2\sin(t) + \frac{2}{3}\sin(t)$$

```
>
```